

French Tutor

Michael Quigley

"French Tutor" is a helpful study aid designed for students who are learning or strengthening French vocabulary and translation skills. For the VIC or 64.

With two children studying French in elementary school, one of my reasons for buying the VIC-20 was to create some French instructional programs. "French Tutor" was suggested by Steve Steinberg's "Language Lab" (*COMPUTE!*, July 1982), which provided for both vocabulary drill and translation practice.

Dummy Words and Custom Accents

It was relatively easy to adapt to the VIC, with a few minor modifications. For example, a dummy word (XX) is needed as the last item in the DATA statements to prevent the program from running out of DATA if a particular word is not in the list.

Another modification involves the use of accents, which Language Lab did not include. The solution was to create the accents with programmable characters, as described in "Custom Characters for the VIC" by David Malmberg (*COMPUTE!'s First Book of VIC*).

In addition to the familiar accented vowels, this program includes some which are used less frequently—the umlauted *e* (as in Noël), *u* (as in Saül), and *o* (for words of German origin). Also included are the combined *æ* for words like *œuvre* and *æ* as in *Cæsar*.

Language Drills

The program is made up of four sections: a French-to-English vocabulary drill, an English-to-French vocabulary drill, a French-to-English translator, and an English-to-French translator. Because of the VIC's memory restraints, there are only 101 words (most of which employ accents and are no longer than five letters). With memory expansion, that total could be increased, although that will necessitate relocating the programmable characters if more than 8K of additional RAM is added.

If you do have more memory and decide to expand the list of words, make the corresponding change in line 43 of Program 2 (line 1610 in Program 3 for the 64), the line that randomly selects the words. If desired, the selection process can be changed so that words will not be repeated.

Typing In the Programs

Since Program 2 almost fills the unexpanded VIC, do not include unnecessary spaces when typing it in. In addition, in order to make some lines fit, you will have to use abbreviated keywords. PRINT becomes *?*, GOSUB becomes GO followed by SHIFT-S, DATA becomes D SHIFT-A, and so on. In particular, lines 1, 3, 4, 5, 60 and 62 require abbreviations.

Adding Words

The maximum number of words allowed for each vocabulary drill is nine. This is because the computer recognizes only the first integer with the GET A\$ statement in line 10, which doesn't require the user to hit the RETURN key. In other words, if you should type in 20 words, the computer would see it as two words. In order to increase that number to 10 or more, eliminate the question mark from line 9. Also, delete lines 10-11 and 18-20 and replace them with the following:

```
10 INPUTN:IFN<1THEN10
11 IFCO=NTHEN14
```

```
18 INPUTN:IFN<1THEN10
20 IFCO=NTHEN14
```

That will allow numbers larger than nine to be used, but you will have to press RETURN after the number is typed in.

For the 64 version, delete the question mark and (MAX. 9) from line 1270. In addition, delete lines 1280-1290 and 1360-1380 and replace them as follows:

```
1280 INPUTN:IFN<1THEN1280
1290 IFCO=NTHEN1320
```

```
1360 INPUTN:IFN<1THEN1280
1380 IFCO=NTHEN1320
```

VIC French Tutor is in two parts. Program 1 will automatically load and run Program 2. Tape users should type in and save Program 1, then type in and save Program 2 as the next program on the tape. To make the VIC version work with a disk

drive, give Program 2 the name "F". Then, in Program 1, delete line 555 and make the following changes:

```
390 IFA$="N"THENPOKE36869,255:GOTO560      :rem 163
580 POKE7993,34:POKE7994,6:POKE7995,34:POKE7996,44
      :POKE7997,56                          :rem 121
590 POKE198,1:POKE631,131:END               :rem 161
```

Program 1. VIC French Tutor, Part 1: Redefined Characters

For error-free program entry, be sure to use "The Automatic Proofreader," Appendix C.

```
10 POKE36879,237                          :rem 105
20 PRINT"{CLR}{4 RIGHT}{4 DOWN}{BLK}"CHR$(122)
                                           :rem 245
30 FORT=1TO8                               :rem 231
40 PRINTTAB(4)"{BLK}"CHR$(125)"{BLU}{RVS}
   {4 SPACES}{WHT}{4 SPACES}{RED}{4 SPACES}"
                                           :rem 161
50 NEXT                                    :rem 164
60 PRINTTAB(4)"{BLK}"CHR$(125)            :rem 13
70 PRINTTAB(4)"{BLK}"CHR$(125)            :rem 14
80 S1=36876:V=36878:POKEV,10              :rem 96
90 READN,D                                :rem 67
100 IFN=-2THEN170                          :rem 209
110 POKES1,(ABS(N))                        :rem 55
120 FORT=1TO(ABS(D))                      :rem 155
130 NEXTT                                  :rem 39
140 POKES1,0                              :rem 164
150 FORN=1TO20:NEXTN                      :rem 5
160 GOTO90                                 :rem 57
170 FORT=1TO2000:NEXTT                    :rem 115
180 GOTO210                               :rem 101
190 DATA-201,-125,-201,-187,-201,-62,-215,-250,-21
   5,-250,-219,-250,-219,-250,-228,-375 :rem 112
200 DATA-223,-125,-215,-1000,-2          :rem 247
210 POKE36879,26:PRINT"{CLR}{BLK}{9 DOWN}"TAB(6)"V
   IC FRENCH":PRINTTAB(7)"TUTORIAL"      :rem 70
220 PRINTTAB(7)"[8 T]"                   :rem 220
230 PRINT"{7 DOWN}{RIGHT}DEFINING CHARACTERS"
                                           :rem 30
260 X=PEEK(56)-2:POKE52,X:POKE56,X:POKE51,PEEK(55)
      :CLR                                  :rem 15
270 CS=256*PEEK(52)+PEEK(51)              :rem 23
280 FORI=CSTOCS+511:POKEI,PEEK(I+32768-CS):NEXT
                                           :rem 166
290 READX                                  :rem 15
300 IFX=-1THEN370                          :rem 222
310 IFX<0THEN290                          :rem 177
320 FORI=XTOX+7:READJ                     :rem 87
```

```
330 IFJ<0THEN320                          :rem 159
340 POKEI,J:NEXT                          :rem 254
350 GOTO290                               :rem 108
370 PRINT"{CLR}{BLK}{10 DOWN}"SPC(5)"INSTRUCTIONS?"
      ":PRINT:PRINTSPC(7)"{RED}Y{BLK}ES OR {RED}N
      {BLK}O"                             :rem 163
380 GETA$:IFA$<>"N"ANDA$<>"Y"THEN380      :rem 51
390 IFA$="N"THENPOKE36869,255:GOTO555     :rem 167
400 PRINT"{CLR}{BLK}{6 DOWN}IN ORDER TO CREATE","F
      RENCH ACCENTS IN","THIS PROGRAM, CERTAIN"
                                           :rem 8
410 PRINT"LETTERS HAVE BEEN","RE-DEFINED USING","P
      ROGRAMMABLE","CHARACTERS."          :rem 215
420 PRINT:PRINT"PRESS {RED}C{BLK} TO CONTINUE."
                                           :rem 224
430 GETA$:IFA$<>"C"THEN430                 :rem 209
440 PRINT"{CLR}{DOWN}{BLK}THE FRENCH CHARACTERS":P
      RINT"ARE {BLU}BLUE{BLK}, THEIR VIC"; :rem 147
450 PRINT"{3 SPACES}EQUIVALENTS {GRN}GREEN{BLK}. "
      :PRINT                               :rem 207
460 POKE36869,255                        :rem 161
470 PRINTTAB(4)"{BLU}< {BLK}{RVS}= {GRN}<{OFF}"SPC
      (4)"{BLU}# {BLK}{RVS}= {GRN}#"      :rem 69
480 PRINT:PRINTTAB(4)"{BLU}$ {BLK}{RVS}= {GRN}$
      {OFF}"SPC(4)"{BLU}% {BLK}{RVS}= {GRN}%"
                                           :rem 155
490 PRINT:PRINTTAB(4)"{BLU}& {BLK}{RVS}= {GRN}&
      {OFF}"SPC(4)"{BLU}+ {BLK}{RVS}= {GRN}+"
                                           :rem 172
500 PRINT:PRINTTAB(4)"{BLU}£ {BLK}{RVS}= {GRN}£
      {OFF}"SPC(4)"{BLU}@ {BLK}{RVS}= {GRN}@":rem 58
510 PRINT:PRINTTAB(4)"{BLU}* {BLK}{RVS}= {GRN}*
      {OFF}"SPC(4)"{BLU}† {BLK}{RVS}= {GRN}†":rem 19
520 PRINT:PRINTTAB(4)"{BLU}[ {BLK}{RVS}= {GRN}[
      {OFF}"SPC(4)"{BLU}] {BLK}{RVS}= {GRN}]"
                                           :rem 116
530 PRINT:PRINTTAB(4)"{BLU}= {BLK}{RVS}= {GRN}=
      {OFF}"SPC(4)"{BLU}< {BLK}{RVS}= {GRN}<"
                                           :rem 247
535 PRINT:PRINTTAB(4)"{BLU}> {BLK}{RVS}= {GRN}>
      {OFF}"SPC(4)"{BLU}/ {BLK}{RVS}= {GRN}/"
                                           :rem 228
540 PRINT:PRINT"{BLK}PRESS {RED}C{BLK} TO CONTINUE
      ."                                   :rem 115
550 GETA$:IFA$<>"C"THEN550                 :rem 215
555 POKE36879,237                        :rem 167
560 PRINT"{CLR}{8 DOWN}{2 RIGHT}{BLK}ONE MOMENT PL
      EASE."                               :rem 235
570 PRINT:PRINTTAB(6)"UN MOMENT,":PRINTTAB(3)"S'IL
      VOUS PLA{T.WHT}"                   :rem 180
```

```

580 POKE198,5:POKE631,78:POKE632,69:POKE633,87:POK
    E634,13:POKE635,131:END :rem 27
600 DATA7168,8,16,126,64,126,64,126,0 :rem 84
620 DATA7384,24,36,0,60,66,66,60,0 :rem 187
630 DATA7392,28,34,64,64,34,28,8,16 :rem 253
640 DATA7400,8,20,0,62,8,8,62,0 :rem 25
650 DATA7408,16,8,126,64,126,64,126,0 :rem 86
660 DATA7416,30,40,72,78,72,40,30,0 :rem 231
670 DATA7448,16,8,66,66,66,66,60,0 :rem 211
680 DATA7456,24,36,0,66,66,66,60,0 :rem 199
690 DATA7464,36,0,60,66,66,66,60,0 :rem 199
695 DATA7472,30,40,72,126,72,72,78,0 :rem 44
700 DATA7504,16,8,60,66,126,66,66,0 :rem 243
720 DATA7512,24,36,0,60,66,126,66,0 :rem 232
725 DATA7544,36,0,66,66,66,66,60,0 :rem 203
730 DATA7648,36,0,126,64,126,64,126,0 :rem 85
740 DATA7656,24,36,126,64,126,64,126,0 :rem 139
750 DATA7664,20,0,62,8,8,8,62,0 :rem 39
760 DATA-1 :rem 21

```

Program 2. VIC French Tutor, Part 2: Vocabulary Drill and Translator

For error-free program entry, be sure to use "The Automatic Proofreader," Appendix C.

```

1 POKE36879,27:PRINT "{CLR}":L$="FRENCH":K$="ENGLIS
  H":R$="{23 SPACES}":POKE36869,240 :rem 37
2 PRINT "{CLR}{BLU}{RVS}"R$"ENTER DESIRED NUMBER"R$
  :rem 62
3 PRINT:PRINT "{GRN}1){BLK} "L$" TO "K$SPC(5)"VOCAB
  ULARY DRILL":PRINT:PRINT "{GRN}2) {BLK}"K$" TO "L
  $SPC(5)"VOCABULARY": :rem 60
4 PRINT "DRILL":PRINT:PRINT "{GRN}3) {BLK}"L$" TO "
  K$SPC(5)"TRANSLATOR":PRINT:PRINT "{GRN}4) {BLK}"K
  $" TO "L$SPC(5)"TRANSLATOR" :rem 53
5 PRINT:PRINT:PRINT "{3 RIGHT}{RVS}{CYN}PUSH {RED}R
  ETURN{CYN} AFTER"SPC(5)" EACH WORD INPUT "SPC(5)
  " IN THIS PROGRAM {OFF}" :rem 159
6 GETMQ$:IFVAL(MQ$)<1ORVAL(MQ$)>4THEN6 :rem 114
8 POKE36869,255:ONVAL(MQ$)GOTO9,18,27,35 :rem 60
9 CO=0:SC=0:PRINT "{CLR}{BLK}{DOWN}HOW MANY WORDS?"
  :rem 165
10 GETA$:IFVAL(A$)<1THEN10 :rem 5
11 LETN=VAL(A$):IFCO=NTHEN14 :rem 229
12 PRINT "{CLR}{DOWN}{GRN}TRANSLATE":PRINT "INTO "K$
  ; :rem 147
13 PRINT "{HOME}{2 DOWN}{RED}"TAB(13)CO+1"{LEFT} OF
  "N":GOSUB43 :rem 229
14 IFCO=NTHEN GOSUB44:GOTO9 :rem 212
15 PRINT "{HOME}{BLK}{5 DOWN}"W$;:INPUTT$ :rem 39

```

```

16 IFT$=E$THENSEC=SC+1:CO=CO+1:PRINT "{BLU}CORRECT":
  FORL=1TO1500:NEXTL:GOTO11 :rem 25
17 IFT$<>E$THENCO=CO+1:PRINT "{BLU}WRONG! IT'S
  {PUR}";E$:FORT=1TO1500:NEXTT:GOTO11 :rem 91
18 CO=0:SC=0:PRINT "{CLR}{BLK}{DOWN}HOW MANY WORDS?"
  :rem 213
19 GETA$:IFVAL(A$)<1THEN19 :rem 23
20 LETN=VAL(A$):IFCO=NTHEN23 :rem 229
21 PRINT "{CLR}{DOWN}{GRN}TRANSLATE":PRINT "INTO "L$
  ; :rem 148
22 PRINT "{HOME}{2 DOWN}{RED}"TAB(13)CO+1"{LEFT} OF
  "N":GOSUB43 :rem 229
23 IFCO=NTHEN GOSUB44:GOTO18 :rem 4
24 PRINT "{HOME}{BLK}{5 DOWN}"E$;:INPUTT$ :rem 21
25 IFT$=W$THENSEC=SC+1:CO=CO+1:PRINT "{BLU}CORRECT":
  FORL=1TO1500:NEXTL:GOTO20 :rem 43
26 IFT$<>W$THENCO=CO+1:PRINT "{BLU}WRONG! IT'S
  {PUR}";W$:FORT=1TO1500:NEXTT:GOTO20 :rem 127
27 PRINT "{CLR}{DOWN}{BLK}ENTER "L$" WORD{5 SPACES}
  OR {GRN}M{BLK} TO GO TO MENU" :rem 137
28 X$="XX":PRINT "{BLK}":INPUTT$ :rem 244
29 IFT$="M"THENRUN :rem 100
30 READE$,W$ :rem 143
31 IFW$=T$THENPRINT "{2 RIGHT}{BLU}"E$:FORT=1TO1500
  :NEXTT:PRINT:RESTORE:GOTO27 :rem 255
32 IFW$=X$THENPRINT "{2 RIGHT}{BLU}TRY AGAIN.":REST
  ORE:GOTO28 :rem 73
33 IFE$<>T$THEN30 :rem 226
34 PRINT "{BLU}{2 RIGHT}TRY AGAIN.":RESTORE:GOTO28
  :rem 89
35 PRINT "{CLR}{DOWN}{BLK}ENTER "K$" WORD{4 SPACES}
  OR {GRN}M{BLK} TO GO TO MENU" :rem 135
36 X$="XX":PRINT "{BLK}":INPUTT$ :rem 243
37 IFT$="M"THENRUN :rem 99
38 READE$,W$ :rem 151
39 IFE$=T$THENPRINT "{2 RIGHT}{BLU}"W$:FORT=1TO1500
  :NEXTT:PRINT:RESTORE:GOTO35 :rem 6
40 IFE$=X$THENPRINT "{2 RIGHT}{BLU}TRY AGAIN.":REST
  ORE:GOTO36 :rem 53
41 IFW$<>T$THEN38 :rem 251
42 PRINT "{BLU}{2 RIGHT}TRY AGAIN.":RESTORE:GOTO36
  :rem 87
43 X=INT(RND(1)*101)+1:RESTORE:FORM=1TOX:READE$,W$
  :NEXTM:RETURN :rem 49
44 PRINT:PRINT "{CLR}{BLK}{4 DOWN}OUT OF";N;"WORDS
  {SPACE}YOU {3 RIGHT}HAVE CORRECTLY"SPC(8)"TRANS
  LATED";SC;"{LEFT}." :rem 122
45 PRINT:PRINT "YOUR SCORE IS";INT((SC/N)*100);SPC(
  6)"PER CENT." :rem 41

```



```

46 PRINT:PRINT"GO AGAIN?":PRINT"{DOWN}{2 RIGHT}
   {RED}Y{BLK} - YES":PRINT"{DOWN}{2 RIGHT}{RED}M
   {SPACE}{BLK}- RETURN TO MENU" :rem 228
47 GETQ$:IFQ$<>"Y"ANDQ$<>"M"THEN47 :rem 2
48 IFQ$="Y"THENRETURN :rem 89
49 IFQ$="M"THENRUN :rem 99
51 DATASUMMER,@T@,APPLE,POMME,HERE,ICI,THERE,L*,NE
   ST,NID :rem 165
52 DATAHOUSE,MAISON,FARM,FERME,WHERE,O#,SAME,M=ME,
   BOX,BO]TE,FRENCH,FRANFAIS,CAKE,G+TEAU :rem 7
53 DATACOW,VACHE,HORSE,CHEVAL,BIRD,OISEAU,CHRISTMA
   S,NO<L,EGG,<UF,EYE,<IL,WORK,<UVRE :rem 132
54 DATACOST,COST,TASTE,GOST,RATHER,PLUT[T,BELIEVE,
   CRO]RE,HEAD,T=TE,BEAST,B=TE,KEY,CL@ :rem 154
55 DATANAME,NOM,YES,OUI,NO,NON,NOSE,NEZ,COFFEE,CAF
   @,BOY,GARFON,DAY,JOUR,CASTLE,CH+TEAU :rem 217
56 DATABLACK,NOIR,BLUE,BLEU,RED,ROUGE,GREEN,VERT,W
   HITE,BLANC,PURPLE,VIOLET,YELLOW,JAUNE :rem 144
57 DATAFEBRUARY,F@VRIER,KNOT,N<UD,TASK,T+CHE,PUPI
   L,@L↑VE,PASTE,P+TE,FOREST,FOR=T,OR,OU :rem 100
58 DATACHOIR,CH<UR,BONE,OS,BEAR,OURS,GOAT,CH↑VRE,C
   ITY,CIT@,NUT,NOIX,MOON,LUNE,BEEF,B<UF :rem 148
59 DATAFATHER,P↑RE,MOTHER,M↑RE,BABY,B@B@,FAIRY,F@E
   ,IRON,FER,FIRE,FEU,WINDOW,FEN=TRE :rem 48
60 DATARULE,R↑GLE,RICE,RIZ,CORN,MA>S,MASTER,MA]TRE
   ,WHEAT,BL@,VERY,TR↑S,SOON,T[T,WINE,VIN :rem 37
61 DATALIFE,VIE,JUNE,JUIN,TAIL,QUEUE,FOOT,PIED,ARM
   ,BRAS,WORD,MOT,LEG,JAMBE,CHILD,ENFANT :rem 21
62 DATASTRONG,FORT,BUILD,B+TIR,AT,*,SWORD,@P@E,FIN
   GER,DOIGT,HEART,C<UR,SKY,CIEL,BEAK,BEC :rem 61
63 DATAHOUSE,MAISON,DOOR,PORTE,SOAP,SAVON,CUT,COUP
   ,LIP,L↑VRE,SCHOOL,@COLE,SUN,SOLEIL :rem 10
64 DATAMILK,LAIT,TEA,TH@,WATER,EAU,ARROW,FL↑CHE,EN
   D,FIN,AUNT,TANTE,TOOTH,DENT,XX,XX :rem 79

```

Program 3. French Tutor, 64 Version

For error-free program entry, be sure to use "The Automatic Proofreader," Appendix C.

```

100 DATA SUMMER,@T@,APPLE,POMME,HERE,ICI,THERE,L*,
   NEST,NID :rem 208
110 DATAHOUSE,MAISON,FARM,FERME,WHERE,O#,SAME,M=ME
   ,BOX,BO]TE :rem 183
120 DATAFRENCH,FRANFAIS,CAKE,G+TEAU :rem 252
130 DATACOW,VACHE,HORSE,CHEVAL,BIRD,OISEAU,CHRISTM
   AS,NO<L :rem 32
140 DATAEGG,<UF,EYE,<IL,WORK,<UVRE :rem 19
150 DATACOST,COST,TASTE,GOST,RATHER,PLUT[T,BELIEVE
   ,CRO]RE :rem 68
160 DATAHEAD,T=TE,BEAST,B=TE,KEY,CL@ :rem 8

```

```

170 DATANAME,NOM,YES,OUI,NO,NON,NOSE,NEZ,COFFEE,CA
   F@,BOY,GARFON :rem 152
180 DATA DAY,JOUR,CASTLE,CH+TEAU :rem 246
190 DATABLACK,NOIR,BLUE,BLEU,RED,ROUGE,GREEN,VERT,
   WHITE,BLANC :rem 21
200 DATAPURPLE,VIOLET,YELLOW,JAUNE :rem 42
210 DATAFEBRUARY,F@VRIER,KNOT,N<UD,TASK,T+CHE,PUPI
   L,@L↑VE :rem 98
220 DATAPASTE,P+TE,FOREST,FOR=T,OR,OU :rem 171
230 DATACHOIR,CH<UR,BONE,OS,BEAR,OURS,GOAT,CH↑VRE
   :rem 19
240 DATACITY,CIT@,NUT,NOIX,MOON,LUNE,BEEF,B<UF
   :rem 45
250 DATAFATHER,P↑RE,MOTHER,M↑RE,BABY,B@B@,FAIRY,F@
   E,IRON :rem 193
260 DATAFER,FIRE,FEU,WINDOW,FEN=TRE :rem 30
270 DATARULE,R↑GLE,RICE,RIZ,CORN,MA>S,MASTER,MA]TR
   E,WHEAT,BL@ :rem 81
280 DATAVERY,TR↑S,SOON,T[T,WINE,VIN :rem 143
290 DATALIFE,VIE,JUNE,JUIN,TAIL,QUEUE,FOOT,PIED,AR
   M,BRAS :rem 190
300 DATAWORD,MOT,LEG,JAMBE,CHILD,ENFANT :rem 12
310 DATASTRONG,FORT,BUILD,B+TIR,AT,*,SWORD,@P@E,FI
   NGER,DOIGT :rem 179
320 DATAHEART,C<UR,SKY,CIEL,BEAK,BEC :rem 57
330 DATAHOUSE,MAISON,DOOR,PORTE,SOAP,SAVON,CUT,COU
   P,LIP,L↑VRE :rem 158
340 DATASCHOOL,@COLE,SUN,SOLEIL :rem 30
350 DATAMILK,LAIT,TEA,TH@,WATER,EAU,ARROW,FL↑CHE,E
   ND,FIN,AUNT,TANTE :rem 180
360 DATATOOTH,DENT,XX,XX :rem 80
365 POKE53280,6 :rem 49
370 POKE53281,12:CH=54272:FORT=CHTOCH+24:POKET,0:N
   EXT:PRINTCHR$(142) :rem 159
375 POKECH+24,15 :rem 124
380 POKECH+5,17:POKECH+6,241:POKECH,100 :rem 18
390 PRINT"{CLR}{8 RIGHT}{6 DOWN}{BLK}"CHR$(122)
   :rem 197
400 FORT=1TO10 :rem 65
410 PRINTTAB(8)"{BLK}"CHR$(125)"{BLU}{RVS}
   {8 SPACES}{WHT}{8 SPACES}{RED}{8 SPACES}"
   :rem 214
420 NEXT :rem 213
430 PRINTTAB(8)"{BLK}"CHR$(125) :rem 66
440 PRINTTAB(8)"{BLK}"CHR$(125) :rem 67
450 FORT=0TO1STEP 0:READA$:IF A$="XX"THENT=1:READ
   A$ :rem 71
451 NEXT :rem 217
460 READN,D :rem 116
470 IFD=-2THEN560 :rem 212

```

```

480 POKECH+1,(ABS(N)):POKECH+4,33      :rem 137
490 FORT=1TO(ABS(D)):NEXT              :rem 30
510 POKECH+4,32                        :rem 64
520 FORN=1TO20:NEXTN                  :rem 6
530 GOTO460                            :rem 107
560 PRINTCHR$(14):GOTO590              :rem 240
570 DATA-16,-125,-16,-187,-16,-62,-22,-250,-22,-25
    0,-25,-250,-25,-250,-33,-375      :rem 230
580 DATA-28,-93,-22,-375,-2,-2       :rem 222
590 PRINT"{CLR}{BLK}{11 DOWN}"TAB(15)"64 FRENCH
    {DOWN}":PRINTTAB(15)"TUTORIAL"    :rem 93
600 PRINTTAB(15)"[8 T]"                :rem 13
603 IFPEEK(12288)=8THENFORII=1TO2500:NEXT:GOTO740
    :rem 31
605 PRINT"{3 DOWN}{3 RIGHT}PLEASE WAIT...DEFINING
    {SPACE}CHARACTERS"                :rem 19
610 PRINT:PRINT                        :rem 235
620 X=48:POKE56,X                     :rem 241
630 CS=12288                           :rem 113
640 POKE56334,PEEK(56334)AND254:POKE1,PEEK(1)AND25
    1                                   :rem 187
650 FORI=CSTOCS+4095:POKEI,PEEK(I+40960):NEXT
    :rem 24
660 READX                             :rem 16
670 IFX=-1THEN740                     :rem 233
680 IFX<0THEN660                      :rem 188
690 FORI=XTOX+7:READJ                 :rem 97
700 IFJ<0THEN690                      :rem 170
710 POKEI+5120,J:NEXT                 :rem 242
720 POKE1,PEEK(1)OR4:POKE56334,PEEK(56334)OR1
    :rem 136
730 GOTO660                           :rem 111
740 PRINT"{CLR}{BLK}{10 DOWN}"SPC(13)"INSTRUCTIONS
    ?{DOWN}":PRINTSPC(15)"{RVS}Y{OFF}ES OR {RVS}N
    {OFF}O"                             :rem 188
750 GETA$:IFA$<>"N"ANDA$<>"Y"THEN750   :rem 53
760 IFA$="N"THEN1150                  :rem 86
770 PRINT"{CLR}{BLK}{6 DOWN}{2 RIGHT}IN ORDER TO C
    REATE FRENCH ACCENTS IN{DOWN}"     :rem 123
780 PRINT"{8 RIGHT}THIS PROGRAM, CERTAIN{DOWN}"
    :rem 235
790 PRINT"{2 RIGHT}LETTERS HAVE BEEN RE-DEFINED US
    ING{DOWN}"                          :rem 86
800 PRINT"{8 RIGHT}PROGRAMMABLE CHARACTERS.{DOWN}"
    :rem 233
810 PRINT"{10 RIGHT}{5 DOWN}PRESS {RVS}C{OFF} TO C
    ONTINUE."                          :rem 139
820 GETA$:IFA$<>"C"THEN820              :rem 215
830 PRINT"{CLR}{DOWN}{4 RIGHT}{BLK}THE FRENCH CHAR
    ACTERS ARE BLACK,"                :rem 236

```

```

840 PRINT"{4 RIGHT}THEIR 64 EQUIVALENTS ARE {WHT}W
    HITE{BLK}.{2 DOWN}"                :rem 86
850 POKE53272,29                      :rem 102
860 PRINTTAB(12)"< {RVS}={OFF} {RVS}{WHT}<{OFF}
    {BLK}"SPC(4)"# {RVS}={OFF} {RVS}{WHT}#{OFF}
    {BLK}"                             :rem 225
870 PRINT:PRINTTAB(12)"$ {RVS}={OFF} {RVS}{WHT}$
    {OFF}{BLK}"SPC(4)"% {RVS}={OFF} {RVS}{WHT}%
    {OFF}{BLK}"                         :rem 55
880 PRINT:PRINTTAB(12)"& {RVS}={OFF} {RVS}{WHT}&
    {OFF}{BLK}"SPC(4)"& {RVS}={OFF} {RVS}{WHT}&
    {OFF}{BLK}"                         :rem 72
890 PRINT:PRINTTAB(12)"£ {RVS}={OFF} {RVS}{WHT}£
    {OFF}{BLK}"SPC(4)"@ {RVS}={OFF} {RVS}{WHT}@
    {OFF}{BLK}"                         :rem 223
900 PRINT:PRINTTAB(12)"* {RVS}={OFF} {RVS}{WHT}*
    {OFF}{BLK}"SPC(4)"↑ {RVS}={OFF} {RVS}{WHT}↑
    {OFF}{BLK}"                         :rem 175
910 PRINT:PRINTTAB(12)"[ {RVS}={OFF} {RVS}{WHT}[
    {OFF}{BLK}"SPC(4)" ] {RVS}={OFF} {RVS}{WHT}]
    {OFF}{BLK}"                         :rem 16
920 PRINT:PRINTTAB(12)"= {RVS}={OFF} {RVS}{WHT}=
    {OFF}{BLK}"SPC(4)"< {RVS}={OFF} {RVS}{WHT}<
    {OFF}{BLK}"                         :rem 147
930 PRINT:PRINTTAB(12)"> {RVS}={OFF} {RVS}{WHT}>
    {OFF}{BLK}"SPC(4)"/ {RVS}={OFF} {RVS}{WHT}/
    {OFF}{BLK}"                         :rem 124
940 PRINTTAB(9)"{2 DOWN}{BLK}PRESS {RVS}C{OFF} TO
    {SPACE}CONTINUE."                 :rem 43
950 GETA$:IFA$<>"C"THEN950              :rem 223
952 DATA7168,8,16,126,64,126,64,126,0
    :rem 94
960 DATA7384,24,36,0,60,66,66,60,0
    :rem 194
965 DATA7392,28,34,64,64,34,28,8,16
    :rem 8
970 DATA7400,8,20,0,62,8,8,62,0
    :rem 31
975 DATA7408,16,8,126,64,126,64,126,0
    :rem 96
980 DATA7416,30,40,72,78,72,40,30,0
    :rem 236
985 DATA7448,16,8,66,66,66,66,60,0
    :rem 220
990 DATA7456,24,36,0,66,66,66,60,0
    :rem 203
995 DATA7464,36,0,60,66,66,66,60,0
    :rem 207
1000 DATA7472,30,40,72,126,72,72,78,0
    :rem 73
1005 DATA7504,16,8,60,66,126,66,66,0
    :rem 34
1015 DATA7512,24,36,0,60,66,126,66,0
    :rem 22
1020 DATA7544,36,0,66,66,66,66,60,0
    :rem 240
1025 DATA7648,36,0,126,64,126,64,126,0
    :rem 131
1030 DATA7656,24,36,126,64,126,64,126,0
    :rem 180
1035 DATA7664,34,0,62,8,8,8,62,0
    :rem 89
1040 DATA-1                           :rem 61
1150 POKE53272,29:CLR:RESTORE         :rem 9
1160 PRINT"{CLR}":L$="FRENCH":K$="ENGLISH"
    :rem 65
1165 R$="{23 SPACES}"                 :rem 196

```

```

1170 POKE53280,7 :rem 93
1180 PRINT"{CLR}{DOWN}{WHT}{RVS}"SPC(10)"ENTER DES
      IRED NUMBER" :rem 121
1190 PRINT"{2 DOWN}{WHT}1){BLK}"L$" TO "K$" {WHT}
      -{BLK} VOCABULARY DRILL" :rem 232
1200 PRINT"{2 DOWN}{WHT}2){BLK}"K$" TO "L$" {WHT}
      -{BLK} VOCABULARY DRILL" :rem 225
1210 PRINT"{2 DOWN}{WHT}3){BLK}"L$" TO "K$" {WHT}
      -{BLK} TRANSLATOR" :rem 126
1220 PRINT"{2 DOWN}{WHT}4){BLK}"K$" TO "L$" {WHT}
      -{BLK} TRANSLATOR" :rem 128
1225 PRINT"{2 DOWN}{WHT}5){BLK}END THE PROGRAM"
      :rem 128
1230 PRINT"{4 DOWN}{3 RIGHT}{RVS}{WHT}PUSH {BLK}RE
      TURN{WHT} AFTER EACH WORD INPUT{DOWN}";
      :rem 153
1240 PRINTSPC(16)"IN THIS PROGRAM" :rem 29
1250 GETMQ$:IFVAL(MQ$)<1ORVAL(MQ$)>5THEN1250
      :rem 151
1255 IFVAL(MQ$)=5THENSYS2048 :rem 192
1260 POKE53272,29:ONVAL(MQ$)GOTO1270,1360,1450,153
      0 :rem 67
1270 CO=0:SC=0:PRINT"{CLR}{BLK}{3 DOWN}{4 RIGHT}HO
      W MANY WORDS? (MAX. 9)" :rem 106
1280 GETA$:IFVAL(A$)<1THEN1280 :rem 217
1290 N=VAL(A$):IFCO=NTHEN1320 :rem 203
1300 PRINT"{CLR}{4 DOWN}{3 RIGHT}TRANSLATE":PRINT"
      {3 RIGHT}INTO "K$ :rem 124
1310 PRINT"{2 UP}"TAB(24)CO+1"{LEFT} OF"N:GOSUB161
      0 :rem 122
1320 IFCO=NTHENGOSUB1620:GOTO1270 :rem 39
1330 PRINT"{3 DOWN}{7 RIGHT}"W$;:INPUTT$ :rem 142
1340 IFT$=E$THENSC=SC+1:CO=CO+1:PRINT"{7 RIGHT}
      {2 DOWN}{WHT}CORRECT 1{BLK}":FORL=1TO2E2:NEXT
      L :rem 12
1345 IFT$=E$THENT$="":GOTO1290 :rem 225
1350 CO=CO+1:PRINT"{7 RIGHT}{2 DOWN}{WHT}WRONG 1
      {2 SPACES}IT'S {BLK}";E$:FORT=1TO1500:NEXTT:G
      OTO1290 :rem 212
1360 CO=0:SC=0:PRINT"{CLR}{BLK}{3 DOWN}{4 RIGHT}HO
      W MANY WORDS?(SHIFT-SPACE)(MAX. 9)" :rem 10
1370 GETA$:IFVAL(A$)<1THEN1370 :rem 217
1380 LETN=VAL(A$):IFCO=NTHEN1410 :rem 176
1390 PRINT"{CLR}{4 DOWN}{3 RIGHT}TRANSLATE":PRINT"
      {3 RIGHT}INTO "L$; :rem 193
1400 PRINT"{UP}"TAB(24)CO+1"{LEFT} OF"N:GOSUB1610
      :rem 233
1410 IFCO=NTHENGOSUB1620:GOTO1360 :rem 39

```

```

1420 PRINT"{3 DOWN}{7 RIGHT}"E$;:INPUTT$ :rem 124
1430 IFT$=W$THENSC=SC+1:CO=CO+1:PRINT"{WHT}
      {2 DOWN}{7 RIGHT}CORRECT 1{BLK}":FORL=1TO1500
      :NEXTL :rem 59
1435 IFT$=W$THEN1380 :rem 135
1440 CO=CO+1:PRINT"{WHT}{2 DOWN}{7 RIGHT}WRONG 1 I
      T'S {BLK}";W$:FORT=1TO1500:NEXTT:GOTO1380
      :rem 230
1450 PRINT"{CLR}{DOWN}{RIGHT}{BLK}ENTER "L$" WORD
      {SPACE}OR {RVS}M{OFF} TO GO TO MENU" :rem 253
1460 X$="XX":PRINT"{DOWN}{RIGHT}";:INPUTT$ :rem 46
1470 IFT$="M"THEN1160 :rem 152
1480 READE$,W$ :rem 249
1490 IFW$=T$THENPRINT"{DOWN}{2 RIGHT}"E$:FORT=1TO1
      500:NEXTT:PRINT:RESTORE:GOTO1450 :rem 188
1500 IFW$=X$THENPRINT"{DOWN}{2 RIGHT}TRY AGAIN.":R
      ESTORE:GOTO1460 :rem 253
1510 IFE$<>T$THEN1480 :rem 173
1520 PRINT"{DOWN}{2 RIGHT}TRY AGAIN.":RESTORE:GOTO
      1460 :rem 13
1530 PRINT"{CLR}{DOWN}{RIGHT}{BLK}ENTER "K$" WORD
      {SPACE}OR {RVS}M{OFF} TO GO TO MENU" :rem 251
1540 X$="XX":PRINT"{DOWN}{RIGHT}";:INPUTT$ :rem 45
1550 IFT$="M"THEN1170 :rem 152
1560 READE$,W$ :rem 248
1570 IFE$=T$THENPRINT"{DOWN}{2 RIGHT}"W$:FORT=1TO1
      500:NEXTT:PRINT:RESTORE:GOTO1530 :rem 186
1580 IFE$=X$THENPRINT"{DOWN}{2 RIGHT}TRY AGAIN.":R
      ESTORE:GOTO1540 :rem 242
1590 IFW$<>T$THEN1560 :rem 198
1600 PRINT"{DOWN}{2 RIGHT}TRY AGAIN.":RESTORE:GOTO
      1540 :rem 11
1610 X=INT(RND(1)*101)+1:RESTORE:FORM=1TOX:READE$,
      W$:NEXTM:RETURN :rem 146
1620 PRINT:PRINT"{CLR}{BLK}{4 DOWN}{3 RIGHT}OUT OF
      ";N;"WORDS, YOU HAVE CORRECTLY" :rem 71
1625 PRINT"{12 RIGHT}TRANSLATED";SC;"{LEFT}."
      :rem 8
1630 PRINT:PRINT"{6 RIGHT}YOUR SCORE IS";INT((SC/N
      )*100);"PER CENT." :rem 203
1640 PRINTTAB(15)"{2 DOWN}GO AGAIN?" :rem 129
1645 PRINT"{DOWN}"TAB(16)"{RVS}Y{OFF} - YES":PRINT
      "{DOWN}"TAB(11)"{RVS}M{OFF} - RETURN TO MENU"
      :rem 96
1650 GETQ$:IFQ$<>"Y"ANDQ$<>"M"THEN1650 :rem 196
1660 IFQ$="Y"THENRETURN :rem 186
1670 IFQ$="M"THEN1170 :rem 152

```